# Maintainer: Konstantin Gizdov <arch at kge dot pw>

_pkgbase='keras'
pkgbase="python-${_pkgbase}"
pkgname=("${pkgbase}")
pkgdesc='A deep learning API written in Python, running on top of the machine learning platform TensorFlow'
url='https://keras.io/'
license=('Apache')
pkgver=2.12.0
pkgrel=2
arch=('any')
source=("https://files.pythonhosted.org/packages/py2.py3/${_pkgbase::1}/${_pkgbase}/${_pkgbase//-/_}-${pkgver}-py2.py3-none-any.whl"
        "${_pkgbase}-${pkgver}-LICENSE::https://raw.githubusercontent.com/keras-team/${_pkgbase}/v${pkgver}/LICENSE")
makedepends=('python-build' 'python-installer' 'python-wheel')
depends=('absl-py' 'python' 'python-h5py' 'python-keras-preprocessing' 'python-numpy' 'python-pandas' 'python-pillow'
         'python-pydot' 'python-scipy' 'python-six' 'python-tensorflow' 'python-yaml' 'tensorboard')
b2sums=('c42f3f1319bb313de0aabe30e2c858cf9355f47dcdd5723eba4c029d5ab9ed283b513e6fa31ae95e94b1028af330248022683b53f83e67154823950bf808936e'
        'dc6395f606b09f8a2fa6e8d28f8436a9b0d2ee7e43b4033542c55eb1bf26e9e6c01fd53770e825b9e996ef15fd2eb77f1e0524d4fc1a3e8bf52d72de3adbd653')

package () {
  python -m installer --destdir="$pkgdir" *.whl
  install -Dm 644 "${_pkgbase}-${pkgver}-LICENSE" "${pkgdir}/usr/share/licenses/${pkgname}/LICENSE"
}
